Mini-Lab 07

We use Laplace transforms to investigate how (electrical) circuits response to perturbations.

1 Heaviside functions, t-shifting

Definition. *The Heaviside (step) function* $u(t-a)$*is defined as*

$$u(t-a)=\begin{cases} 0 & \text{if } t<a \\ 1 & \text{if } t>a \end{cases}. \ (a \geqslant 0)$$

The Laplace transform is

$$\mathcal{L}[u(t-a)] = \int_0^\infty e^{-st} u(t-a)\,\mathrm{d}t = \int_a^\infty e^{-st}\,\mathrm{d}t = \left[-\frac{e^{-st}}{s}\right]_{t=a}^{t\to\infty} = \frac{e^{-as}}{s}$$

Theorem. *(Frequency-shifting) If* $F=\mathcal{L}(f)$*, then the frequency-shifted function* $F(s-a)$ *is the Laplace transform of* $\tilde{f}(t)=e^{at}f(t-a)$

$$\mathcal{L}(\tilde{f})(s)=\mathcal{L}(e^{at}f(t-a))(s)=F(s-a).$$

Theorem. *(Time-shifting) If* $F=\mathcal{L}(f)$*, then the time-shifted function* $\tilde{f}(t)=f(t-a)u(t-a)$ *has Laplace transform*

$$\mathcal{L}(\tilde{f})(s)=\mathcal{L}(f(t-a)u(t-a))(s)=e^{-as}F(s).$$

2 Circuit response to perturbations

2.1 Switch circuit on over time interval

A perturbation of amplitude V_0 applied for time interval $t\in[a,b]$ (e.g., voltage for electrical circuits) is

$$v(t)=V_0[u(t-a)-u(t-b)].$$

```
In[2]:= v[t_,V0_,a_,b_] := V0 (UnitStep[t-a]-UnitStep[t-b]);
        vplt=Plot[v[t,1,1,2],{t,0,3},Axes->False,Frame->True,FrameLabel->{"t","v(t)"}];
        Export["/home/student/courses/MATH528/vplt.png",vplt]
```

/home/student/courses/MATH528/vplt.png

```
In[3]:=
```



Figure 1. Single rectangular wave

2.1.1 $R-C$ (resistance-storage) circuit

- Model

$$Ry(t)+\frac{q(t)}{K}=Ry(t)+\frac{1}{K}\int_0^t y(\tau)\,\mathrm{d}\tau=v(t)=V_0[u(t-a)-u(t-b)]$$

```
In[28]:= tmodel = R y[t] + 1/K Integrate[y[tau],{tau,0,t}] == v[t,V0,a,b]
```

$$\frac{\int_0^t y(\tau)\ d\tau}{K}+R\,y(t)=\mathrm{V0}(\theta\,(t-a)-\theta\,(t-b))$$

```
In[29]:=
```

- Laplace transform

$$RY(s)+\frac{Y(s)}{sK}=\frac{V_0}{s}[e^{-as}-e^{-bs}]$$

```
In[29]:= smodel = Simplify[LaplaceTransform[tmodel,t,s],a>0 && b>0] /.
         LaplaceTransform[y[t], t, s] -> Y[s]
```

$$\frac{\mathrm{V0}(e^{-bs}-e^{-as})+Y(s)\left(\frac{1}{K}+R\,s\right)}{s}=0$$

```
In[30]:=
```

- Solution in frequency space

$$I(s)=\frac{V_0/R}{s+(RK)^{-1}}[e^{-as}-e^{-bs}]=F(s)[e^{-as}-e^{-bs}]$$

```
In[32]:= sol[s_]=Simplify[Y[s] /. Solve[smodel,Y[s]][[1,1]]]
```

$$\frac{K\,\mathrm{V0}(e^{-as}-e^{-bs})}{K\,R\,s+1}$$

```
In[33]:= InputForm[sol[s]]
```

((E^(-(a*s)) - E^(-(b*s)))*K*V0)/(1 + K*R*s)

```
In[34]:=
```

- Response function in time

$$f(t) = \mathcal{L}^{-1}(F)(t) = \frac{V_0}{R} e^{-t/(RC)}$$

```
In[34]:= F[s_]=(K*V0)/(1 + K*R*s)
```

$$\frac{K\text{V0}}{K R s + 1}$$

```
In[35]:= f[t_]=InverseLaplaceTransform[F[s],s,t]
```

$$\frac{\text{V0}e^{-\frac{t}{KR}}}{R}$$

```
In[36]:=
```

- Response to single rectangular wave from frequency shifting

$$i(t) = f(t-a)u(t-a) - f(t-b)u(t-b)$$

```
In[42]:= y[t_,V0_,R_,K_,a_,b_] = f[t-a] UnitStep[t-a] - f[t-b] UnitStep[t-b]
```

$$\frac{\text{V0}\theta(t-a)\, e^{-\frac{t-a}{KR}}}{R} - \frac{\text{V0}\theta(t-b)\, e^{-\frac{t-b}{KR}}}{R}$$

```
In[43]:= yi[t_,V0_,R_,K_,a_,b_]=InverseLaplaceTransform[sol[s],s,t]
```

$$\frac{\text{V0}e^{-\frac{t}{KR}}\left(\theta(t-a)\, e^{\frac{a}{KR}} - \theta(t-b)\, e^{\frac{b}{KR}}\right)}{R}$$

```
In[44]:= params = {V0->1,R->1,K->1,a->1,b->2};
         solplt = Plot[{y[t,V0,R,K,a,b],yi[t,V0,R,K,a,b]} /. params,{t,0,4},Axes->False,
         Frame->True,FrameLabel->{"t","v(t)"}];
         Export["/home/student/courses/MATH528/solplt.png",solplt]
```

/home/student/courses/MATH528/solplt.png

```
In[45]:=
```



Figure 2. Circuit response to single rectangular wave